Roy Smart

Solar Flares & CMEs

Problem Set 3

In[1]:=
```
Clear["Global`*"]
```

Part a.

Define array to store numerical values of constants

In[2]:=
```
vals = {};
```

The problem statement provides the following expression for the radiative loss function

In[3]:=
```
Λ = Λ0 (10^-6 T)^α;
$Assumptions = T > 0 && Λ0 > 0 && α ∈ Reals;
AppendTo[vals,
  Λ0 → Quantity[1.2 × 10^-22, "Ergs" * "Centimeters"^3 * "Seconds"^-1 * "Kelvins"^1/2]];
```

In Lecture 9, the radiative cooling time is defined as

In[6]:=
```
(τrad = (3 k T)/(n Λ)) // Framed
$Assumptions = $Assumptions && k > 0 && n > 0;
```

Out[6]=
$$\boxed{\frac{3 \times 1000000^{\alpha}\, k\, T^{1-\alpha}}{n\, \Lambda0}}$$

Part b.

Also in Lecture 9, we are given an expression for the conductive cooling time as

In[8]:=
```
τcond = (21 k n L^2)/(8 κ0 T^(5/2));
$Assumptions = $Assumptions && L > 0 && κ0 > 0 ;
AppendTo[vals,
  κ0 → Quantity[10^-6, "Ergs" / ("Seconds" * "Centimeters" * "Kelvins"^(7/2))]];
AppendTo[vals, k → 1.38 × 10^-16 ergs/K ];
```

Set the two cooling times equal and solve for the electron number density

In[12]:=
```
(nb = Part[Solve[τrad == τcond, n] // FullSimplify, 2, 1, 2]) // Framed
```

Out[12]= $\boxed{\dfrac{2^{\frac{3}{2}+3\alpha} \times 125^{\alpha}\, T^{\frac{7}{4}-\frac{\alpha}{2}} \sqrt{\frac{\kappa 0}{\Lambda 0}}}{\sqrt{7}\, L}}$

Solve for where the derivative with respect to temperature of the above expression is positive as a function of α.

In[13]:=
```
(αb = Reduce[D[nb, T] > 0, α] // Simplify) // Framed // Quiet
$Assumptions = $Assumptions && αb;
```

Out[13]= $\boxed{2\alpha < 7}$

Part c.

Solve Equation 2 in the problem statement for number density. This is the peak number density

In[15]:=
```
nc = ε/(3 L k T);
$Assumptions = $Assumptions && ε > 0;
```

Again, solve the condition where the two cooling times are equal for the peak temperature, with the number density replaced with the above expression

In[17]:=
```
eqc = τrad == τcond /. n → nc // FullSimplify;
(Tstar = Part[Solve[eqc, T] // FullSimplify, 1, 1, 2] // Quiet) // Framed
```

Out[18]=
$$\boxed{\left(\frac{9}{7}\right)^{\frac{2}{-11+2\alpha}}\left(\frac{8^{1+2\alpha}\times 15\,625^{\alpha}\,k^2\,\kappa 0}{\varepsilon^2\,\Lambda 0}\right)^{\frac{2}{-11+2\alpha}}}$$

So then the peak number density is found by plugging the peak temperature back into the expression for peak number density

In[19]:=
```
(nstar = nc /. T → Tstar // FullSimplify) // Framed
```

Out[19]=
$$\boxed{\frac{3^{-1-\frac{4}{-11+2\alpha}}\times 7^{\frac{2}{-11+2\alpha}}\,\varepsilon\left(\frac{8^{1+2\alpha}\times 15\,625^{\alpha}\,k^2\,\kappa 0}{\varepsilon^2\,\Lambda 0}\right)^{-\frac{2}{-11+2\alpha}}}{k\,L}}$$

Part d.

Equation 3 in the problem statement is a differential equation in pressure and time

In[20]:=
```
p = 2 n k T /. n → nb /. T → T[t];
rhs = -5 n^2 Λ /. n → nb /. T → T[t];
oded = 3/2 D[p, t] == rhs
```

Out[22]=
$$\frac{3\times 2^{\frac{3}{2}+3\alpha}\times 125^{\alpha}\,k\left(\frac{11}{4}-\frac{\alpha}{2}\right)\sqrt{\frac{\kappa 0}{\Lambda 0}}\,T[t]^{\frac{7}{4}-\frac{\alpha}{2}}\,T'[t]}{\sqrt{7}\,L} == -\frac{40\,\kappa 0\,T[t]^{7/2}}{7\,L^2}$$

Solve the differential equation and write in the correct form

In[23]:=
```
$Assumptions = $Assumptions && Ts > 0 && t ∈ Reals;
Td1 = Part[DSolve[{oded, T[0] == Ts}, T[t], t] // Quiet, 1, 1, 2];
Td2 = Td1;
Td2[[2, 1, 2, 2]] = Td1[[2, 1, 2, 2]] (Ts / Tstar)^(1/Td1[[2,2]]);
Td3 = Td2 // PowerExpand // Simplify // PowerExpand // Simplify;
Td3[[4, 1]] = Td3[[4, 1]] // Apart;
Td4 = Td3;
Td4[[4, 1]] = Td3[[4, 1]] / Td3[[4, 1, 1]] // Apart;
Td5 = Td4 Td3[[4, 1, 1]]^Td3[[4,2]];
posCond = Reduce[-Td5[[5, 2]] > 0];
$Assumptions = $Assumptions && posCond;
(Td6 = (Td5 / Td5[[-1]] // Simplify) Td5[[-1]]) // Framed
```

Out[34]=

$$\boxed{Ts \left(1-\left(2^{-2+\frac{34}{11-2\alpha}-\frac{52\alpha}{11-2\alpha}}\times 3^{-1+\frac{3}{-11+2\alpha}+\frac{2\alpha}{-11+2\alpha}}\times 5^{\frac{11}{11-2\alpha}-\frac{44\alpha}{11-2\alpha}}\times 7^{\frac{4}{-11+2\alpha}-\frac{2\alpha}{-11+2\alpha}}\,k^{\frac{14}{-11+2\alpha}}\,t\,(3+2\alpha)\,\varepsilon^{\frac{3}{11-2\alpha}+\frac{2\alpha}{11-2\alpha}}\,\kappa 0^{-\frac{4}{-11+2\alpha}+\frac{2\alpha}{-11+2\alpha}}\,\Lambda 0^{\frac{7}{11-2\alpha}}\right)\Big/\left(L\,(-11+2\alpha)\right)\right)^{-\frac{4}{3+2\alpha}}}$$

Write in terms of ω and μ

In[35]:=
```
ωd = Td6[[2, 1, 2]] / t;
μd = -Td6[[2, 2]];
$Assumptions = $Assumptions && ω > 0 && μ > 0;
Td = Td6 /. ωd → ω /. -μd → -μ
```

Out[38]= $Ts\,(1+t\,\omega)^{-\mu}$

The solution will tend to zero if μ is positive

In[39]:=
```
posCond // Framed
```

Out[39]= $\boxed{\alpha > -\frac{3}{2}}$

Part e.

The contribution function is approximated as

In[40]:=
```
Gλ = Gλ0 Exp[-Log[T / Tλ]^2 / σλ^2];
$Assumptions = $Assumptions && Tλ > 0 && σλ > 0 && Gλ0 > 0;
```

The emissivity is related to the contribution function through

In[42]:=
```
ελ = n^2 Gλ /. n → nb
```

Out[42]=
$$\frac{2^{3+6\alpha} \times 125^{2\alpha}\, e^{-\frac{\mathrm{Log}\left[\frac{T}{T\lambda}\right]^2}{\sigma\lambda^2}}\, G\lambda 0\, T^{\frac{7}{2}-\alpha}\, \kappa 0}{7\, L^2\, \Lambda 0}$$

Take the derivative with respect to temperature to find the maximum emissivity

In[43]:=
```
(Tλpk = Part[Solve[D[ελ, T] == 0, T], 1, 1, 2] // FullSimplify) // Framed
```

Out[43]=
$$\boxed{e^{\frac{1}{4}(7-2\alpha)\,\sigma\lambda^2}\, T\lambda}$$

Find whether β is greater or less than unity

In[44]:=
```
re = Tλpk[[1]] > 1
```

Out[44]=
$$e^{\frac{1}{4}(7-2\alpha)\,\sigma\lambda^2} > 1$$

In[45]:=
```
Reduce[re] // FullSimplify // Framed
```

Out[45]=
$$\boxed{\mathrm{True}}$$

Therefore β is greater than unity.

Part f.

Find the time at which maximum emissivity is reached

In[46]:=
```
Tλpk == Td
```

Out[46]=
$$e^{\frac{1}{4}(7-2\alpha)\,\sigma\lambda^2}\, T\lambda == Ts\,(1 + t\,\omega)^{-\mu}$$

In[47]:=
```
(tλ = Part[Solve[Tλpk == Td, t] // FullSimplify // Quiet, 1, 1, 2]) // Framed
```

Out[47]=
$$\boxed{\frac{-1 + \left(\frac{e^{\frac{1}{4}(7-2\alpha)\,\sigma\lambda^2}\, T\lambda}{Ts}\right)^{-1/\mu}}{\omega}}$$

The lifetime of the emission of the spectral line is defined through the relation

In[48]:=
```
ελf = ελ /. T → Td;
fr = 1/ελf D[D[ελf, t], t] == - 2/Δτλ^2 /. t → tλ // FullSimplify
```

Out[49]= $\left(\frac{e^{\frac{1}{4}(7-2\alpha)\,\sigma\lambda^2}\,T\lambda}{Ts}\right)^{2/\mu} \Delta\tau\lambda\,\mu^2\,\omega^2 == \frac{\sigma\lambda^2}{\Delta\tau\lambda}$

Solve for $\Delta\tau\lambda$

In[50]:=
```
(Δτλf = Part[Solve[fr, Δτλ], 2, 1, 2]) // Framed
```

Out[50]= $\boxed{\frac{\left(\frac{e^{\frac{1}{4}(7-2\alpha)\,\sigma\lambda^2}\,T\lambda}{Ts}\right)^{-1/\mu}\sigma\lambda}{\mu\,\omega}}$

Part g.

Plug in the values given in Table 1 in the problem statement

In[51]:=
```
FeXXI = {Gλ0 → 1.64 × 10^-25 ergs * cm^3/s , Tλ → Quantity[11.51 × 10^6, "Kelvins"], σλ → 0.3};
FeXVIII = {Gλ0 → 1.43 × 10^-25 ergs * cm^3/s , Tλ → Quantity[6.91 × 10^6, "Kelvins"], σλ → 0.44};
FeXIV = {Gλ0 → 6.13 × 10^-25 ergs * cm^3/s , Tλ → Quantity[1.99 × 10^6, "Kelvins"], σλ → 0.25};
FeIX = {Gλ0 → 37.84 × 10^-25 ergs * cm^3/s , Tλ → Quantity[0.82 × 10^6, "Kelvins"], σλ → 0.42};
```

Save the other given numerical values to memory

In[55]:=
```
AppendTo[vals, L → Quantity[5 × 10^9, "Centimeters"]];
AppendTo[vals, ε → Quantity[2 × 10^12, "Ergs" * "Centimeters"^-2]];
AppendTo[vals, α → -1 / 2];
```

Evaluate the time of peak emission

In[58]:=
```
tλ /. Ts → Tstar /. μ → μd /. ω → ωd /. vals /. FeXXI // N // Framed
tλ /. Ts → Tstar /. μ → μd /. ω → ωd /. vals /. FeXVIII // N // Framed
tλ /. Ts → Tstar /. μ → μd /. ω → ωd /. vals /. FeXIV // N // Framed
tλ /. Ts → Tstar /. μ → μd /. ω → ωd /. vals /. FeIX // N // Framed
```

Out[58]= $\boxed{-161.537\ \mathrm{s}}$

Out[59]= $\boxed{134.086\ \mathrm{s}}$

Out[60]= $\boxed{2498.32\ \mathrm{s}}$

Out[61]= $\boxed{4240.92\ \mathrm{s}}$

Evaluate the duration of the emission

In[62]:=
```
Δτλf /. Ts → Tstar /. μ → μd /. ω → ωd /. vals /. FeXXI // N // Framed
Δτλf /. Ts → Tstar /. μ → μd /. ω → ωd /. vals /. FeXVIII // N // Framed
Δτλf /. Ts → Tstar /. μ → μd /. ω → ωd /. vals /. FeXIV // N // Framed
Δτλf /. Ts → Tstar /. μ → μd /. ω → ωd /. vals /. FeIX // N // Framed
```

Out[62]= $\boxed{271.039\ \mathrm{s}}$

Out[63]= $\boxed{462.561\ \mathrm{s}}$

Out[64]= $\boxed{558.348\ \mathrm{s}}$

Out[65]= $\boxed{1303.97\ \mathrm{s}}$

In[66]:=
```
ωd /. vals
```

Out[66]= 0.00050801 per second