Solar Flares & CMEs
Problem Set 2
Roy Smart

```
Clear["Global`*"]
```

Problem 1

Part a.

Declare the assumptions for this problem

```
$Assumptions = λ > 0 && z > 0 && b > 0 && c > 0 && Ics > 0 && h > 0 && a > 0;
```

The flux function for this problem is given by

$$A[y\_, z\_] = \lambda\,\mathrm{ArcTan}\left[\frac{y+b}{z}\right] - \lambda\,\mathrm{ArcTan}\left[\frac{y-b}{z}\right] + \frac{\mathrm{Ics}}{c}\,\mathrm{Log}\left[\frac{y^2 + (z+h)^2}{y^2 + (z-h)^2}\right];$$

Plot this function to ensure that it has the correct behaviour

```
ContourPlot[A[y, z] /. {b → 1, λ → 1, h → 1, Ics → 0.5, c → 1},
 {y, -1, 1}, {z, 0, 2}, Contours → 50]
```



The reconnection flux is measured between the origin and the X-point. Find the X-point by setting the magnetic field on the y-axis equal to zero and solve for the height z. Find the magnetic field by taking a curl.

```
B = Cross[Grad[A[y, z], {x, y, z}], {1, 0, 0}] /. y → 0 // FullSimplify
```

$$\left\{0, \frac{4\,h\,Ics}{c\,h^2 - c\,z^2} - \frac{2\,b\,\lambda}{b^2 + z^2}, 0\right\}$$

Solve for the height of the X-point

```
zx = Part[Solve[B[[2]] == 0, z] // FullSimplify, 2, 1, 2]
```

$$\sqrt{\frac{b\,h\,(-2\,b\,Ics + c\,h\,\lambda)}{2\,h\,Ics + b\,c\,\lambda}}$$

Expand this solution for a high-lying rope (h >> b)

```
zxa =
 Normal[Series[zx // FullSimplify, {h, Infinity, 0}]] /. λ → 2 Ics / c // FullSimplify
```

$$\sqrt{b\,h}$$

Plug this expression into the flux function and continue the approximation

```
(ψr = Series[A[0, zxa], {h, Infinity, 1}] /. λ → 2 Ics / c // Normal) // FullSimplify //
 Framed
```

$$\boxed{\frac{8\sqrt{\frac{b}{h}}\,\text{Ics}}{c}}$$

Part b.

The measured CME velocity is given as

```
v = v0/2 (Tanh[(t - 2 τ)/τ] + 1) /. v0 → b/τ
```

$$\frac{b\left(1+\text{Tanh}\left[\frac{t-2\tau}{\tau}\right]\right)}{2\tau}$$

Height as a function of time is found by performing an indefinite integral over the given velocity

```
h[t_] = Integrate[v, t] + const
```

$$\text{const}+\frac{b\,t}{2\tau}+\frac{1}{2}\,b\,\text{Log}\left[\text{Cosh}\left[\frac{t-2\tau}{\tau}\right]\right]$$

```
cv = Part[Solve[h[2 τ] == 5 b, const], 1]
```

$$\{\text{const}\to 4\,b\}$$

```
h[t_] = h[t] /. cv
```

$$4\,b+\frac{b\,t}{2\tau}+\frac{1}{2}\,b\,\text{Log}\left[\text{Cosh}\left[\frac{t-2\tau}{\tau}\right]\right]$$

So the flux function becomes

```
ψ[t_] = ψr /. h → h[t] /. cv // FullSimplify
```

$$\frac{8\sqrt{2}\,\text{Ics}\sqrt{\frac{\tau}{t+8\tau+\tau\,\text{Log}\left[\text{Cosh}\left[2-\frac{t}{\tau}\right]\right]}}}{c}$$

Plot the resulting flux function along with the velocity for τ=50sec

```
rules50 = {τ → 50, Ics → 3 × 10^21, b → 6 × 10^9, c → 3 × 10^10};
Plot[{ψ[t] / 10^3, v} /. rules50, {t, 0, 1000}, PlotRange → {{0, 1000}, {0, 4 × 10^8}}]
```



Plot the resulting flux function along with the velocity for τ=100sec

```
rules150 = {τ → 150, Ics → 3 × 10^21, b → 6 × 10^9, c → 3 × 10^10};
Plot[{ψ[t] / 10^3, v} /. rules150 , {t, 0, 1000}, PlotRange → {{0, 1000}, {0, 4 × 10^8}}]
```



Part c.

Find the reconnection rate

```
dψdt = D[ψ[t], t] // FullSimplify
```

$$\frac{1}{c\,\tau} 4\sqrt{2}\ \mathrm{Ics}\left(\frac{\tau}{t + 8\tau + \tau\,\mathrm{Log}\left[\mathrm{Cosh}\left[2 - \frac{t}{\tau}\right]\right]}\right)^{3/2}\left(-1 + \mathrm{Tanh}\left[2 - \frac{t}{\tau}\right]\right)$$

Find the acceleration

```
ah = D[v, t] // FullSimplify
```

$$\frac{b\,\mathrm{Sech}\left[2-\frac{t}{\tau}\right]^2}{2\,\tau^2}$$

Plot the reconnection rate and acceleration for τ=50 sec

```
Plot[{dψdt/10^3, ah} /. rules50, {t, 0, 1000},
 PlotRange → {{0, 1000}, {-6 × 10^5, 1.3 × 10^6}}]
```



Plot the reconnection rate and acceleration for τ=150 sec

```
Plot[{dψdt/10^3, ah} /. rules150, {t, 0, 1000},
 PlotRange → {{0, 1000}, {-6 × 10^5, 1.3 × 10^6}}]
```



Find the time of maximum reconnection rate for τ=50 sec

```
(dψdtmin50 = Part[FindMinimum[dψdt /. rules50, {t, 300}], 2, 1, 2]) // Framed
```

$\boxed{150.843}$

Find the time of maximum acceleration for τ=50 sec

```
(ahmax50 = Part[FindMaximum[ah /. rules50, {t, 100}] // Quiet, 2, 1, 2]) // Framed
```

```
100.
```

Find the time of maximum reconnection rate for τ=150 sec

```
(dψdtmin150 = Part[FindMinimum[dψdt /. rules150, {t, 300}], 2, 1, 2]) // Framed
```

```
452.528
```

Find the time of maximum acceleration for τ=150 sec

```
(ahmax150 = Part[FindMaximum[ah /. rules150, {t, 100}] // Quiet, 2, 1, 2]) // Framed
```

```
300.
```

Part d.

Express the volume in terms of an initial volume V0 and a change in volume ΔV

```
V = V0 + ΔV;
```

Denote the initial volume in terms of an area a and length L

```
V0 = a h0
```

```
a h0
```

The change in volume is said to be proportional to the height h

```
ΔV = a (h - h0);
```

The number density is the number of particles over volume

$$n = \frac{N}{V};$$

$$n0 = \frac{N}{V0};$$

The ratio of the number density to the initial number density is then

```
(nR = n / n0 // FullSimplify) // Framed
```

$$\boxed{\frac{h0}{h}}$$

Plot this ratio as a function of time

```
Plot[nR /. {h → h[t], h0 → h[0], L → 1, α → 1} /. rules50, {t, 0, 1000}]
```



The temperature is given by the adiabatic process

```
γ = 5 / 3;
T = β / V^(γ-1);
T0 = β / V0^(γ-1);
```

The ratio of the temperature to the initial temperature is then

```
(TR = T / T0 // FullSimplify) // Framed
```

$$\boxed{\left(\frac{h0}{h}\right)^{2/3}}$$

Plot this ratio as a function of time

```
Plot[TR /. {h → h[t], h0 → h[0], L → 1} /. rules50, {t, 0, 1000}]
```



The emission measure is the square of the number density integrated over the length of the axial flux tube

```
EM = n^2 h // FullSimplify
EM0 = n0^2 h0 // FullSimplify
```

$$\frac{N^2}{a^2\, h}$$

$$\frac{N^2}{a^2\, h0}$$

The ratio of the initial emission measure to the final emission measure is then

```
(EMR = EM / EM0 // FullSimplify) // Framed
```

$$\boxed{\frac{h0}{h}}$$

and the plot of this ratio is

```
Plot[EMR /. {h → h[t], h0 → h[0], L → 1} /. rules50, {t, 0, 1000}]
```



Part e.

The response function for our instrument is given as

```
G[T_] := Exp[-(T - T00)^2/(2 Tw^2)]
```

Evaluate at current temperature

```
GR = G[TR T00] /. T00 → 2 Tw // FullSimplify
```

$e^{-2\left(-1+\left(\frac{h0}{h}\right)^{2/3}\right)^2}$

Plot the counts, which is the product of the temperature sensitivity and the emission measure

```
Plot[GR EMR /. {h → h[t], h0 → h[0], L → 1} /. rules50, {t, 0, 1000}]
```



Evaluate the extent of dimming when the flux rope is twice the initial height

```
GR EMR /. h → 2 h[0] /. h0 → h[0] // FullSimplify // Framed
```

$$\boxed{\frac{1}{2}\, e^{-2\left(-1+\frac{1}{2^{2/3}}\right)^{2}}}$$