PHSX 591 Solar Flares & CMEs

Problem Set 1

Roy Smart

Due Feb. 15th

```
Clear["Global`*"]
```

Problem 1

Part a.

The flux function for Figure 1a is given by

$$Ap = \lambda \operatorname{ArcTan}\left[\frac{y+b}{z}\right] - \lambda \operatorname{ArcTan}\left[\frac{y+a}{z}\right] + \lambda \operatorname{ArcTan}\left[\frac{y-a}{z}\right] - \lambda \operatorname{ArcTan}\left[\frac{y-b}{z}\right];$$

Define the assumptions for this expression

```
$Assumptions = λ > 0 && a > 0 && z > 0 && b > a && c > 0 && Ics < 0
```

```
λ > 0 && a > 0 && z > 0 && b > a && c > 0 && Ics < 0
```

Plot this function to see what it looks like

```
ContourPlot[Ap /. {b → 2, a → 1, λ → 1}, {y, -2, 2}, {z, 0, 2}, Contours → 50]
```



The X point is located where the magnetic field is equal to zero. Therefore, we will begin by taking a derivative of the flux function to determine the magnetic field.

```
Bp = Cross[Grad[Ap, {x, y, z}], {1, 0, 0}] // FullSimplify
```

$$\left\{0,\ \frac{1}{z^2}\left(a\left(\frac{1}{1+\frac{(a-y)^2}{z^2}}+\frac{1}{1+\frac{(a+y)^2}{z^2}}\right)+ y\left(-\frac{1}{1+\frac{(a-y)^2}{z^2}}+\frac{1}{1+\frac{(b-y)^2}{z^2}}+\frac{1}{1+\frac{(a+y)^2}{z^2}}-\frac{1}{1+\frac{(b+y)^2}{z^2}}\right)-\frac{b}{1+\frac{(b-y)^2}{z^2}}-\frac{b}{1+\frac{(b+y)^2}{z^2}}\right)\lambda,\ \frac{1}{z}\left(-\frac{1}{1+\frac{(a-y)^2}{z^2}}+\frac{1}{1+\frac{(b-y)^2}{z^2}}+\frac{1}{1+\frac{(a+y)^2}{z^2}}-\frac{1}{1+\frac{(b+y)^2}{z^2}}\right)\lambda\right\}$$

Evaluate this field at y=0

```
Bpz = Bp /. y → 0
```

$$\left\{0,\ \frac{\left(\frac{2a}{1+\frac{a^2}{z^2}}-\frac{2b}{1+\frac{b^2}{z^2}}\right)\lambda}{z^2},\ 0\right\}$$

Find the height for which all components of the magnetic field are zero by solving the y-component for z

```
sol1aa = Solve[Bpz[[2]] == 0, z] // FullSimplify
```

$$\left\{\left\{z \to -\sqrt{a\,b}\right\},\ \left\{z \to \sqrt{a\,b}\right\}\right\}$$

Select the positive solution, since we are interested in the height of the X point above the photosphere

```
zx = sol1aa[[2, 1, 2]]
```

$\sqrt{a\,b}$

The flux between this X point and the origin is given by evaluating the flux function at this X point.

```
ψp12 = Ap /. {y → 0, z → zx} // FullSimplify
```

$$2\,\lambda\left(-\mathrm{ArcTan}\left[\sqrt{\frac{a}{b}}\right]+\mathrm{ArcTan}\left[\sqrt{\frac{b}{a}}\right]\right)$$

Check limit by expanding about a/b=0

```
Series[ψp12 /. b → 1, {a, 0, 1}] /. a → 0
```

$\pi\,\lambda$

Check another limit by expanding about a/b=1

```
Series[ψp12 /. b → 1, {a, 1, 1}] /. a → 1 // Quiet
```

0

Part b.

The change in flux can be written as

```
Δψp12 = (ψp12 /. a → a + Δa) - ψp12 // FullSimplify
```

$$2\,\lambda\left(\mathrm{ArcTan}\left[\sqrt{\frac{a}{b}}\right]-\mathrm{ArcTan}\left[\sqrt{\frac{b}{a}}\right]+\mathrm{ArcTan}\left[\sqrt{\frac{b}{a+\Delta a}}\right]-\mathrm{ArcTan}\left[\sqrt{\frac{a+\Delta a}{b}}\right]\right)$$

Expand the change in flux about Δa/a=0 to leading order.

```
Δψp12 /. b → 2 a /. a → 1
```

$$2\,\lambda\left(\mathrm{ArcTan}\left[\frac{1}{\sqrt{2}}\right]-\mathrm{ArcTan}\left[\sqrt{2}\right]+\mathrm{ArcTan}\left[\sqrt{2}\,\sqrt{\frac{1}{1+\Delta a}}\right]-\mathrm{ArcTan}\left[\frac{\sqrt{1+\Delta a}}{\sqrt{2}}\right]\right)$$

```
Δψp12 = Normal[Series[Δψp12 /. b → 2 a /. a → 1, {Δa, 0, 1}]] /. Δa → Δa / a
```

$$-\frac{2\sqrt{2}\,\Delta a\,\lambda}{3\,a}$$

Part c.

The potential field after the wire is added is given by

$$A = Ap + \frac{Ics}{c} Log\left[\frac{y^2 + (z+h)^2}{y^2 + (z-h)^2}\right]$$

$$\lambda \, ArcTan\left[\frac{-a+y}{z}\right] - \lambda \, ArcTan\left[\frac{a+y}{z}\right] - \lambda \, ArcTan\left[\frac{-b+y}{z}\right] + \lambda \, ArcTan\left[\frac{b+y}{z}\right] + \frac{Ics \, Log\left[\frac{y^2+(h+z)^2}{y^2+(-h+z)^2}\right]}{c}$$

```
ContourPlot[A /. {b → 2, a → 1, λ → 1, Ics → -0.01, c → 1, h → √2},
 {y, -2, 2}, {z, 0, 3}, Contours → 150]
```



Again, the null points are where the magnetic field is equal to zero. Start by calculating the new magnetic field for this new potential

```
B = Cross[Grad[A, {x, y, z}], {1, 0, 0}] // FullSimplify
```

$$\left\{0, \frac{4\,h\,Ics\,(h^2+y^2-z^2)}{c\,(y^2+(h-z)^2)\,(y^2+(h+z)^2)} + \left(\frac{a-y}{(a-y)^2+z^2} + \frac{-b+y}{(b-y)^2+z^2} + \frac{a+y}{(a+y)^2+z^2} - \frac{b+y}{(b+y)^2+z^2}\right)\lambda,\right.$$
$$\left.\frac{8\,h\,Ics\,y\,z}{c\,(y^2+(h-z)^2)\,(y^2+(h+z)^2)} + z\left(-\frac{1}{(a-y)^2+z^2} + \frac{1}{(b-y)^2+z^2} + \frac{1}{(a+y)^2+z^2} - \frac{1}{(b+y)^2+z^2}\right)\lambda\right\}$$

Evaluate this field at y=0, since we know the nulls are located on the z-axis

```
By = (B /. y → 0)[[2]] // FullSimplify
```

$$\frac{4\,h\,Ics}{c\,h^2 - c\,z^2} + \frac{2\,a\,\lambda}{a^2+z^2} - \frac{2\,b\,\lambda}{b^2+z^2}$$

Now solve for the exact height of the nulls. We will define a quantity $Ir^2 = Ics/c\lambda$ to use for expanding expressions.

```
h = zx;
sol2 = Solve[By == 0, z] ;
zn1 = sol2[[2, 1, 2]] /. Ics → c λ Ir^2 // Simplify
zn2 = sol2[[4, 1, 2]] /. Ics → c λ Ir^2 // Simplify
```

$$\sqrt{\left(\left(a^2 b - a b^2 + \sqrt{a^5 b}\, Ir^2 + \sqrt{a b^5}\, Ir^2 - (a+b)\sqrt{\left(a (a-b) b\, Ir^2 \left(2\sqrt{a b} + a\, Ir^2 - b\, Ir^2\right)\right)}\right) \Big/ \left(a - b - 2\sqrt{a b}\, Ir^2\right)\right)}$$

$$\sqrt{\left(\left(a^2 b - a b^2 + \sqrt{a^5 b}\, Ir^2 + \sqrt{a b^5}\, Ir^2 + (a+b)\sqrt{\left(a (a-b) b\, Ir^2 \left(2\sqrt{a b} + a\, Ir^2 - b\, Ir^2\right)\right)}\right) \Big/ \left(a - b - 2\sqrt{a b}\, Ir^2\right)\right)}$$

Expand the exact expression for the location of the nulls for small current

```
zna1 = Series[zn1, {Ir, 0, 1}] // Normal // Simplify
zna2 = Series[zn2, {Ir, 0, 1}] // Normal // Simplify
```

$$\frac{1}{2}\sqrt{a b}\left(2 - \frac{\sqrt{2}\,(a+b)\, Ir}{\sqrt{a-b}\,(a b)^{1/4}}\right)$$

$$\frac{1}{2}\sqrt{a b}\left(2 + \frac{\sqrt{2}\,(a+b)\, Ir}{\sqrt{a-b}\,(a b)^{1/4}}\right)$$

Write answer in terms of current

```
(z1 = zna1 /. Ir → Sqrt[Ics/(c λ)] // Simplify) // Framed
(z2 = zna2 /. Ir → Sqrt[Ics/(c λ)] // Simplify) // Framed
```

$$\boxed{\frac{1}{2}\sqrt{a b}\left(2 - \frac{\sqrt{2}\,(a+b)\sqrt{\frac{Ics}{a c \lambda - b c \lambda}}}{(a b)^{1/4}}\right)}$$

$$\boxed{\frac{1}{2}\sqrt{a b}\left(2 + \frac{\sqrt{2}\,(a+b)\sqrt{\frac{Ics}{a c \lambda - b c \lambda}}}{(a b)^{1/4}}\right)}$$

Check our answer by evaluating the flux function for both nulls and verifying that they are the same. Start by evaluating the exact expression for the flux function at

the nulls.

```
A1 = A /. Ics → c λ Ir^2 /. y → 0 /. z → zna1 // Simplify
A2 = A /. Ics → c λ Ir^2 /. y → 0 /. z → zna2 // Simplify
```

$$\lambda\left(-2\,\text{ArcTan}\left[\frac{2\,a}{\sqrt{a\,b}\,\left(2-\frac{\sqrt{2}\,(a+b)\,\text{Ir}}{\sqrt{a-b}\,(a\,b)^{1/4}}\right)}\right]+2\,\text{ArcTan}\left[\frac{2\,b}{\sqrt{a\,b}\,\left(2-\frac{\sqrt{2}\,(a+b)\,\text{Ir}}{\sqrt{a-b}\,(a\,b)^{1/4}}\right)}\right]+\text{Ir}^2\,\text{Log}\left[\left(-4\sqrt{a-b}\,(a\,b)^{1/4}+\sqrt{2}\,a\,\text{Ir}+\sqrt{2}\,b\,\text{Ir}\right)^2\Big/\left(2\,(a+b)^2\,\text{Ir}^2\right)\right]\right)$$

$$\lambda\left(-2\,\text{ArcTan}\left[\frac{2\,a}{\sqrt{a\,b}\,\left(2+\frac{\sqrt{2}\,(a+b)\,\text{Ir}}{\sqrt{a-b}\,(a\,b)^{1/4}}\right)}\right]+2\,\text{ArcTan}\left[\frac{2\,b}{\sqrt{a\,b}\,\left(2+\frac{\sqrt{2}\,(a+b)\,\text{Ir}}{\sqrt{a-b}\,(a\,b)^{1/4}}\right)}\right]+\text{Ir}^2\,\text{Log}\left[\left(4\sqrt{a-b}\,(a\,b)^{1/4}+\sqrt{2}\,a\,\text{Ir}+\sqrt{2}\,b\,\text{Ir}\right)^2\Big/\left(2\,(a+b)^2\,\text{Ir}^2\right)\right]\right)$$

next, expand the argument of the logarithm for small current.

```
A1[[2, 3, 2, 1]] = Series[A1[[2, 3, 2, 1]], {Ir, 0, -2}] // Normal;
A2[[2, 3, 2, 1]] = Series[A2[[2, 3, 2, 1]], {Ir, 0, -2}] // Normal;
A1
A2
```

$$\lambda\left(-2\,\text{ArcTan}\left[\frac{2\,a}{\sqrt{a\,b}\,\left(2-\frac{\sqrt{2}\,(a+b)\,\text{Ir}}{\sqrt{a-b}\,(a\,b)^{1/4}}\right)}\right]+2\,\text{ArcTan}\left[\frac{2\,b}{\sqrt{a\,b}\,\left(2-\frac{\sqrt{2}\,(a+b)\,\text{Ir}}{\sqrt{a-b}\,(a\,b)^{1/4}}\right)}\right]+\text{Ir}^2\,\text{Log}\left[\frac{8\,(a-b)\,\sqrt{a\,b}}{(a+b)^2\,\text{Ir}^2}\right]\right)$$

$$\lambda\left(-2\,\text{ArcTan}\left[\frac{2\,a}{\sqrt{a\,b}\,\left(2+\frac{\sqrt{2}\,(a+b)\,\text{Ir}}{\sqrt{a-b}\,(a\,b)^{1/4}}\right)}\right]+2\,\text{ArcTan}\left[\frac{2\,b}{\sqrt{a\,b}\,\left(2+\frac{\sqrt{2}\,(a+b)\,\text{Ir}}{\sqrt{a-b}\,(a\,b)^{1/4}}\right)}\right]+\text{Ir}^2\,\text{Log}\left[\frac{8\,(a-b)\,\sqrt{a\,b}}{(a+b)^2\,\text{Ir}^2}\right]\right)$$

Finally, expand the rest of the expression for small current

```
(ψ12 = Series[A1, {Ir, 0, 2}] // Simplify // Normal) // Framed
(Series[A2, {Ir, 0, 2}] // Simplify // Normal) // Framed
```

$$\boxed{2\,\lambda\left(-\text{ArcTan}\left[\sqrt{\frac{a}{b}}\right]+\text{ArcTan}\left[\sqrt{\frac{b}{a}}\right]\right)+\text{Ir}^2\,\lambda\left(1+\text{Log}\left[\frac{8\,(a-b)\,\sqrt{a\,b}}{(a+b)^2}\right]-2\,\text{Log}[\text{Ir}]\right)}$$

$$\boxed{2\,\lambda\left(-\text{ArcTan}\left[\sqrt{\frac{a}{b}}\right]+\text{ArcTan}\left[\sqrt{\frac{b}{a}}\right]\right)+\text{Ir}^2\,\lambda\left(1+\text{Log}\left[\frac{8\,(a-b)\,\sqrt{a\,b}}{(a+b)^2}\right]-2\,\text{Log}[\text{Ir}]\right)}$$

the flux function for both nulls is the same value to leading order.

Part d.

Compute the flux difference

$$\Delta\psi 12a = \psi 12 - \psi p12 \;/.\; Ir \to \sqrt{\frac{Ics}{c\,\lambda}} \;/.\; b \to 2\,a \;//\; Expand$$

$$\frac{Ics}{c} + \frac{Ics\,Log\left[-\frac{8\sqrt{2}\sqrt{a^2}}{9\,a}\right]}{c} - \frac{2\,Ics\,Log\left[\sqrt{\frac{Ics}{c\,\lambda}}\right]}{c}$$

The logarithm terms provide the largest contribution to the flux since the current is small.

$$(\Delta\psi 12b = \Delta\psi 12a[[2]] + \Delta\psi 12a[[3]] \;//\; Simplify) \;//\; Framed$$

$$\boxed{-\frac{Ics\,Log\left[-\frac{9\,Ics}{8\sqrt{2}\,c\,\lambda}\right]}{c}}$$

Find the length of the current sheet by finding the distance between the nulls

$$\left(L = \frac{1}{2}(z2 - z1) \;/.\; b \to 2\,a \;//\; Simplify\right) \;//\; Framed$$

$$\boxed{\frac{3\,a}{2^{1/4}\sqrt{-\frac{c\,\lambda}{Ics}}}}$$

Part e.

Perform a variable substitution to write given integral in terms of current

$$d\Delta\psi 12 = D[\Delta\psi 12b, Ics]$$

$$-\frac{1}{c} - \frac{Log\left[-\frac{9\,Ics}{8\sqrt{2}\,c\,\lambda}\right]}{c}$$

Perform the integral to find the change in energy

```
(ΔϵM = 1/c ∫Ics dΔψ12 ⅆIcs // Simplify) // Framed
```

$$\boxed{-\frac{\text{Ics}^2\left(1+2\,\text{Log}\left[-\frac{9\,\text{Ics}}{8\sqrt{2}\,c\,\lambda}\right]\right)}{4\,c^2}}$$

Part f.

Ampere's law is given as

```
amp = -4 π Ics/c == 4 L Bi;
```

Solve for the magnetic field

```
Clear[Bi]
sol2 = Solve[amp, Bi];
(Bi = sol2[[1, 1, 2]] // FullSimplify) // Framed
```

$$\boxed{\frac{2^{1/4}\,\pi\,\lambda}{3\,a\sqrt{-\frac{c\,\lambda}{\text{Ics}}}}}$$

Now, the Alven speed is

```
vA = Bi/Sqrt[μ0 ρ0] // FullSimplify;
```

We can use this to find the Alven transit time, given as

```
(τA = 2 L/vA // FullSimplify) // Framed
```

$$\boxed{\frac{9\sqrt{2}\,a^2\sqrt{\mu 0\,\rho 0}}{\pi\,\lambda}}$$

Part g.

Save the provided values to memory

```
values = {a → 3 × 10^9 cm , b → 6 × 10^9 cm , Lx → 10^10 cm ,
   ψSrc → 10^22 Mx , Δa → -10^9 cm , c → 3 × 10^10 cm/s , μ0 → 4 π 10^-7 H/m };
```

Find the value of the parameter λ evaluating the flux function at y=∞

```
(λV = N[ψSrc/(π Lx) /. values]) // Framed
```

$\boxed{3.1831 \times 10^{11}\ \mathrm{Mx/cm}}$

```
AppendTo[values, λ → λV];
```

Solve for the current by equating the change in flux from part b. to that found in part d.

```
Icgs2mks = 10^8/((10 H/m) (3 × 10^10 cm/s));
sol3 = NSolve[Δψp12 == Δψ12b /. values, Ics];
IcsV = Re[sol3[[1, 1, 2]]];
AppendTo[values, Ics → IcsV];
UnitConvert[IcsV Icgs2mks, "amps"] // Framed
```

$\boxed{-4.90637 \times 10^{12}\ \mathrm{A}}$

Now we are free to find a value for the length of the current sheet using the results of part d.

```
L /. values // Framed
```

$\boxed{9.39594 \times 10^{9}\ \mathrm{cm}}$

The height of the current sheet was found in part c.

```
N[h /. values] // Framed
```

$\boxed{4.24264 \times 10^{9}\ \mathrm{cm}}$

and the energy was found in part e.

```
(ΔϵMV = UnitConvert[ΔϵM Lx / μ0 /. values, "ergs"]) // Framed
```

$\boxed{-6.74193 \times 10^{31}\ \mathrm{ergs}}$

Part h.

Save the additional numeric values to memory

```
vribV = vrib → 3 km/s;
Bz0V = Bz0 → 300 G;

AppendTo[values, vribV];
AppendTo[values, Bz0V];
```

The total reconnection time can be found by equating the change in flux to the product of the magnetic field and the change in area. We then solve the expression for time. Note: the factor of 2 is due to the two flare ribbons.

```
(τrx = UnitConvert[Δψ12b/(2 Bz0 vrib) /. values, "seconds"]) // Framed
```

555.751 s

we are then free to compute the average power using the quotient of the total energy release and the total reconnection time

```
(PM = ΔϵMV / τrx) // Framed
```

-1.21312×10^{29} ergs/s

The electric field is the negative change in flux per unit length

```
(EM = UnitConvert[- Δψ12b / τrx /. values, "volts/meter"]) // Framed
```

-180. V/m

The energy flux incident on each ribbon is

```
PM/(2 Lx vrib τrx) /. values // Framed
```

-3.63808×10^{10} ergs/(cm^{2}s)

```
AppendTo[values, ρ0 → 10^-15 g/cm^3];
machnum = UnitConvert[τA /. values, "second"] / τrx // Framed
```

$\boxed{0.023106}$

```
τA /. values
```

$0.000406074\ \sqrt{g}\ \mathrm{cm}\ \sqrt{H}/\mathrm{Mx}$

```
UnitConvert[τA /. values, "second"]
```

12.8412 s