PHSX 591 Solar Flares & CMEs

Problem Set 1
Roy Smart
Due Feb. 15th

```
Clear["Global`*"]
```

Problem 1

Part a.

The flux function for Figure 1a is given by

```
Ap = λ ArcTan[(y + b)/z] - λ ArcTan[(y + a)/z] + λ ArcTan[(y - a)/z] - λ ArcTan[(y - b)/z];
```

$$Ap = \lambda \,\mathrm{ArcTan}\left[\frac{y+b}{z}\right] - \lambda \,\mathrm{ArcTan}\left[\frac{y+a}{z}\right] + \lambda \,\mathrm{ArcTan}\left[\frac{y-a}{z}\right] - \lambda \,\mathrm{ArcTan}\left[\frac{y-b}{z}\right];$$

Define the assumptions for this expression

```
$Assumptions = λ > 0 && a > 0 && z > 0 && b > a && c > 0 && Ics < 0
λ > 0 && a > 0 && z > 0 && b > a && c > 0 && Ics < 0
```

Plot this function to see what it looks like

```
ContourPlot[Ap /. {b → 2, a → 1, λ → 1}, {y, -2, 2}, {z, 0, 2}, Contours → 50]
```



The X point is located where the magnetic field is equal to zero. Therefore, we will begin by taking a derivative of the flux function to determine the magnetic field.

```
Bp = Cross[Grad[Ap, {x, y, z}], {1, 0, 0}] // FullSimplify
```

$$\left\{0, \frac{1}{z^2}\left(a\left(\frac{1}{1+\frac{(a-y)^2}{z^2}}+\frac{1}{1+\frac{(a+y)^2}{z^2}}\right)+ y\left(-\frac{1}{1+\frac{(a-y)^2}{z^2}}+\frac{1}{1+\frac{(b-y)^2}{z^2}}+\frac{1}{1+\frac{(a+y)^2}{z^2}}-\frac{1}{1+\frac{(b+y)^2}{z^2}}\right)-\frac{b}{1+\frac{(b-y)^2}{z^2}}-\frac{b}{1+\frac{(b+y)^2}{z^2}}\right)\lambda, \frac{1}{z}\left(-\frac{1}{1+\frac{(a-y)^2}{z^2}}+\frac{1}{1+\frac{(b-y)^2}{z^2}}+\frac{1}{1+\frac{(a+y)^2}{z^2}}-\frac{1}{1+\frac{(b+y)^2}{z^2}}\right)\lambda\right\}$$

Evaluate this field at y=0

```
Bpz = Bp /. y → 0
```

$$\left\{0, \frac{\left(\frac{2a}{1+\frac{a^2}{z^2}}-\frac{2b}{1+\frac{b^2}{z^2}}\right)\lambda}{z^2}, 0\right\}$$

Find the height for which all components of the magnetic field are zero by solving the y-component for z

```
sol1aa = Solve[Bpz[[2]] == 0, z] // FullSimplify
```

$$\left\{\left\{z \to -\sqrt{a\,b}\right\}, \left\{z \to \sqrt{a\,b}\right\}\right\}$$

Select the positive solution, since we are interested in the height of the X point above the photosphere

```
(zx = sol1aa[[2, 1, 2]]) // Framed
```

$$\boxed{\sqrt{a\,b}}$$

The flux between this X point and the origin is given by evaluating the flux function at this X point.

```
ψp12 = Ap /. {y → 0, z → zx} // FullSimplify
```

$$2\,\lambda\left(-\text{ArcTan}\left[\sqrt{\frac{a}{b}}\right] + \text{ArcTan}\left[\sqrt{\frac{b}{a}}\right]\right)$$

Check limit by expanding about a/b=0

```
Series[ψp12 /. b → 1, {a, 0, 1}] /. a → 0
```

$$\pi\,\lambda$$

Check another limit by expanding about a/b=1

```
Series[ψp12 /. b → 1, {a, 1, 1}] /. a → 1 // Quiet
```

0

Part b.

The change in flux can be written as

```
Δψp12 = (ψp12 /. a → a + Δa) - ψp12 // FullSimplify
```

$$2\,\lambda\left(\text{ArcTan}\left[\sqrt{\frac{a}{b}}\right] - \text{ArcTan}\left[\sqrt{\frac{b}{a}}\right] + \text{ArcTan}\left[\sqrt{\frac{b}{a+\Delta a}}\right] - \text{ArcTan}\left[\sqrt{\frac{a+\Delta a}{b}}\right]\right)$$

Expand the change in flux about Δa/a=0 to leading order.

```
Δψp12 /. b → 2 a /. a → 1
```

$$2\,\lambda\left(\text{ArcTan}\left[\frac{1}{\sqrt{2}}\right] - \text{ArcTan}\left[\sqrt{2}\right] + \text{ArcTan}\left[\sqrt{2}\,\sqrt{\frac{1}{1+\Delta a}}\right] - \text{ArcTan}\left[\frac{\sqrt{1+\Delta a}}{\sqrt{2}}\right]\right)$$

```
(Δψp12 = Normal[Series[Δψp12 /. b → 2 a /. a → 1, {Δa, 0, 1}]] /. Δa → Δa / a) // Framed
```

$$\boxed{-\frac{2\sqrt{2}\,\Delta a\,\lambda}{3\,a}}$$

Part c.

The potential field after the wire is added is given by

```
A = Ap + Ics/c Log[(y^2 + (z + h)^2)/(y^2 + (z - h)^2)]
```

$$\lambda\,\mathrm{ArcTan}\left[\frac{-a+y}{z}\right]-\lambda\,\mathrm{ArcTan}\left[\frac{a+y}{z}\right]-\lambda\,\mathrm{ArcTan}\left[\frac{-b+y}{z}\right]+\lambda\,\mathrm{ArcTan}\left[\frac{b+y}{z}\right]+\frac{\mathrm{Ics}\,\mathrm{Log}\left[\frac{y^2+(h+z)^2}{y^2+(-h+z)^2}\right]}{c}$$

```
ContourPlot[A /. {b → 2, a → 1, λ → 1, Ics → -0.01, c → 1, h → √2},
 {y, -2, 2}, {z, 0, 3}, Contours → 150]
```



Again, the null points are where the magnetic field is equal to zero. Start by calculating the new magnetic field for this new potential

```
B = Cross[Grad[A, {x, y, z}], {1, 0, 0}] // FullSimplify
```

$$\left\{0,\ \frac{4\,h\,\mathrm{Ics}\,(h^2+y^2-z^2)}{c\left(y^2+(h-z)^2\right)\left(y^2+(h+z)^2\right)}+\left(\frac{a-y}{(a-y)^2+z^2}+\frac{-b+y}{(b-y)^2+z^2}+\frac{a+y}{(a+y)^2+z^2}-\frac{b+y}{(b+y)^2+z^2}\right)\lambda,\right.$$
$$\left.\frac{8\,h\,\mathrm{Ics}\,y\,z}{c\left(y^2+(h-z)^2\right)\left(y^2+(h+z)^2\right)}+z\left(-\frac{1}{(a-y)^2+z^2}+\frac{1}{(b-y)^2+z^2}+\frac{1}{(a+y)^2+z^2}-\frac{1}{(b+y)^2+z^2}\right)\lambda\right\}$$

Evaluate this field at y=0, since we know the nulls are located on the z-axis

```
By = (B /. y → 0)[[2]] // FullSimplify
```

$$\frac{4\,h\,\text{Ics}}{c\,h^2 - c\,z^2} + \frac{2\,a\,\lambda}{a^2+z^2} - \frac{2\,b\,\lambda}{b^2+z^2}$$

Now solve for the exact height of the nulls. We will define a quantity $Ir^2 = Ics/c\lambda$ to use for expanding expressions.

```
h = zx;
sol2 = Solve[By == 0, z];
zn1 = sol2[[2, 1, 2]] /. Ics → c λ Ir² // Simplify
zn2 = sol2[[4, 1, 2]] /. Ics → c λ Ir² // Simplify
```

$$\sqrt{\left(\left(a^2 b - a b^2 + \sqrt{a^5 b}\ \text{Ir}^2 + \sqrt{a b^5}\ \text{Ir}^2 - (a+b)\sqrt{\left(a\,(a-b)\,b\,\text{Ir}^2\left(2\sqrt{a\,b} + a\,\text{Ir}^2 - b\,\text{Ir}^2\right)\right)}\right)\Big/\left(a - b - 2\sqrt{a\,b}\ \text{Ir}^2\right)\right)}$$

$$\sqrt{\left(\left(a^2 b - a b^2 + \sqrt{a^5 b}\ \text{Ir}^2 + \sqrt{a b^5}\ \text{Ir}^2 + (a+b)\sqrt{\left(a\,(a-b)\,b\,\text{Ir}^2\left(2\sqrt{a\,b} + a\,\text{Ir}^2 - b\,\text{Ir}^2\right)\right)}\right)\Big/\left(a - b - 2\sqrt{a\,b}\ \text{Ir}^2\right)\right)}$$

Expand the exact expression for the location of the nulls for small current

```
zna1 = Series[zn1, {Ir, 0, 1}] // Normal // Simplify
zna2 = Series[zn2, {Ir, 0, 1}] // Normal // Simplify
```

$$\frac{1}{2}\sqrt{a\,b}\left(2 - \frac{\sqrt{2}\,(a+b)\,\text{Ir}}{\sqrt{a-b}\,(a\,b)^{1/4}}\right)$$

$$\frac{1}{2}\sqrt{a\,b}\left(2 + \frac{\sqrt{2}\,(a+b)\,\text{Ir}}{\sqrt{a-b}\,(a\,b)^{1/4}}\right)$$

Write answer in terms of current

```
(z1 = zna1 /. Ir → Sqrt[Ics/(c λ)] // Simplify) // Framed
(z2 = zna2 /. Ir → Sqrt[Ics/(c λ)] // Simplify) // Framed
```

$$\boxed{\frac{1}{2}\sqrt{a\,b}\left(2-\frac{\sqrt{2}\,(a+b)\sqrt{\frac{Ics}{a\,c\,\lambda-b\,c\,\lambda}}}{(a\,b)^{1/4}}\right)}$$

$$\boxed{\frac{1}{2}\sqrt{a\,b}\left(2+\frac{\sqrt{2}\,(a+b)\sqrt{\frac{Ics}{a\,c\,\lambda-b\,c\,\lambda}}}{(a\,b)^{1/4}}\right)}$$

Check our answer by evaluating the flux function for both nulls and verifying that they are the same. Start by evaluating the exact expression for the flux function at the nulls.

```
A1 = A /. Ics → c λ Ir^2 /. y → 0 /. z → zna1 // Simplify
A2 = A /. Ics → c λ Ir^2 /. y → 0 /. z → zna2 // Simplify
```

$$\lambda\left(-2\,\mathrm{ArcTan}\left[\frac{2\,a}{\sqrt{a\,b}\left(2-\frac{\sqrt{2}\,(a+b)\,Ir}{\sqrt{a-b}\,(a\,b)^{1/4}}\right)}\right]+2\,\mathrm{ArcTan}\left[\frac{2\,b}{\sqrt{a\,b}\left(2-\frac{\sqrt{2}\,(a+b)\,Ir}{\sqrt{a-b}\,(a\,b)^{1/4}}\right)}\right]+Ir^2\,\mathrm{Log}\left[\left(-4\sqrt{a-b}\,(a\,b)^{1/4}+\sqrt{2}\,a\,Ir+\sqrt{2}\,b\,Ir\right)^2\Big/\left(2\,(a+b)^2\,Ir^2\right)\right]\right)$$

$$\lambda\left(-2\,\mathrm{ArcTan}\left[\frac{2\,a}{\sqrt{a\,b}\left(2+\frac{\sqrt{2}\,(a+b)\,Ir}{\sqrt{a-b}\,(a\,b)^{1/4}}\right)}\right]+2\,\mathrm{ArcTan}\left[\frac{2\,b}{\sqrt{a\,b}\left(2+\frac{\sqrt{2}\,(a+b)\,Ir}{\sqrt{a-b}\,(a\,b)^{1/4}}\right)}\right]+Ir^2\,\mathrm{Log}\left[\left(4\sqrt{a-b}\,(a\,b)^{1/4}+\sqrt{2}\,a\,Ir+\sqrt{2}\,b\,Ir\right)^2\Big/\left(2\,(a+b)^2\,Ir^2\right)\right]\right)$$

next, expand the argument of the logarithm for small current.

```
A1[[2, 3, 2, 1]] = Series[A1[[2, 3, 2, 1]], {Ir, 0, -2}] // Normal;
A2[[2, 3, 2, 1]] = Series[A2[[2, 3, 2, 1]], {Ir, 0, -2}] // Normal;
A1
A2
```

$$\lambda\left(-2\,\text{ArcTan}\left[\frac{2\,a}{\sqrt{a\,b}\left(2-\frac{\sqrt{2}\,(a+b)\,\text{Ir}}{\sqrt{a-b}\,(a\,b)^{1/4}}\right)}\right]+2\,\text{ArcTan}\left[\frac{2\,b}{\sqrt{a\,b}\left(2-\frac{\sqrt{2}\,(a+b)\,\text{Ir}}{\sqrt{a-b}\,(a\,b)^{1/4}}\right)}\right]+\text{Ir}^2\,\text{Log}\left[\frac{8\,(a-b)\,\sqrt{a\,b}}{(a+b)^2\,\text{Ir}^2}\right]\right)$$

$$\lambda\left(-2\,\text{ArcTan}\left[\frac{2\,a}{\sqrt{a\,b}\left(2+\frac{\sqrt{2}\,(a+b)\,\text{Ir}}{\sqrt{a-b}\,(a\,b)^{1/4}}\right)}\right]+2\,\text{ArcTan}\left[\frac{2\,b}{\sqrt{a\,b}\left(2+\frac{\sqrt{2}\,(a+b)\,\text{Ir}}{\sqrt{a-b}\,(a\,b)^{1/4}}\right)}\right]+\text{Ir}^2\,\text{Log}\left[\frac{8\,(a-b)\,\sqrt{a\,b}}{(a+b)^2\,\text{Ir}^2}\right]\right)$$

Finally, expand the rest of the expression for small current

```
(ψ12 = Series[A1, {Ir, 0, 2}] // Simplify // Normal) // Framed
(Series[A2, {Ir, 0, 2}] // Simplify // Normal) // Framed
```

$$\boxed{2\,\lambda\left(-\text{ArcTan}\left[\sqrt{\frac{a}{b}}\right]+\text{ArcTan}\left[\sqrt{\frac{b}{a}}\right]\right)+\text{Ir}^2\,\lambda\left(1+\text{Log}\left[\frac{8\,(a-b)\,\sqrt{a\,b}}{(a+b)^2}\right]-2\,\text{Log}[\text{Ir}]\right)}$$

$$\boxed{2\,\lambda\left(-\text{ArcTan}\left[\sqrt{\frac{a}{b}}\right]+\text{ArcTan}\left[\sqrt{\frac{b}{a}}\right]\right)+\text{Ir}^2\,\lambda\left(1+\text{Log}\left[\frac{8\,(a-b)\,\sqrt{a\,b}}{(a+b)^2}\right]-2\,\text{Log}[\text{Ir}]\right)}$$

the flux function for both nulls is the same value to leading order.

Part d.

Compute the flux difference

$$\Delta\psi\text{12a} = \psi\text{12} - \psi\text{p12} \;/.\; \text{Ir} \to \sqrt{\frac{\text{Ics}}{c\,\lambda}} \;/.\; b \to 2\,a \;//\; \text{Expand}$$

$$\frac{\text{Ics}}{c}+\frac{\text{Ics}\,\text{Log}\left[-\frac{8\sqrt{2}\,\sqrt{a^2}}{9\,a}\right]}{c}-\frac{2\,\text{Ics}\,\text{Log}\left[\sqrt{\frac{\text{Ics}}{c\,\lambda}}\right]}{c}$$

The logarithm terms provide the largest contribution to the flux since the current is small.

```
(Δψ12b = Δψ12a[[2]] + Δψ12a[[3]] // Simplify) // Framed
```

$$\boxed{-\frac{\text{Ics}\,\text{Log}\left[-\frac{9\,\text{Ics}}{8\sqrt{2}\,c\,\lambda}\right]}{c}}$$

Find the length of the current sheet by finding the distance between the nulls

```
(L = 1/2 (z2 - z1) /. b → 2 a // Simplify) // Framed
```

$$\boxed{\frac{3\,a}{2^{1/4}\sqrt{-\frac{c\,\lambda}{\text{Ics}}}}}$$

Part e.

Perform a variable substitution to write given integral in terms of current

```
dΔψ12 = D[Δψ12b, Ics]
```

$$-\frac{1}{c}-\frac{\text{Log}\left[-\frac{9\,\text{Ics}}{8\sqrt{2}\,c\,\lambda}\right]}{c}$$

Perform the integral to find the change in energy

```
(ΔϵM = 1/c ∫Ics dΔψ12 ⅆIcs // Simplify) // Framed
```

$$\boxed{-\frac{\text{Ics}^2\left(1+2\,\text{Log}\left[-\frac{9\,\text{Ics}}{8\sqrt{2}\,c\,\lambda}\right]\right)}{4\,c^2}}$$

Part f.

Ampere’s law is given as

```
amp = -(4 π Ics)/c == 4 L Bi;
```

Solve for the magnetic field

```
Clear[Bi]
sol2 = Solve[amp, Bi];
(Bi = sol2[[1, 1, 2]] // FullSimplify) // Framed
```

$$\boxed{\frac{2^{1/4}\,\pi\,\lambda}{3\,a\,\sqrt{-\frac{c\,\lambda}{Ics}}}}$$

Now, the Alven speed is

$$vA = \frac{Bi}{\sqrt{\mu 0\,\rho 0}} \text{ // FullSimplify;}$$

We can use this to find the Alven transit time, given as

$$\left(\tau A = \frac{2\,L}{vA} \text{ // FullSimplify}\right) \text{ // Framed}$$

$$\boxed{\frac{9\,\sqrt{2}\,a^2\,\sqrt{\mu 0\,\rho 0}}{\pi\,\lambda}}$$

Part g.

Save the provided values to memory

$$\text{values} = \{a \to 3\times 10^9\ \text{cm},\ b \to 6\times 10^9\ \text{cm},\ Lx \to 10^{10}\ \text{cm},\ \psi Src \to 10^{22}\ \text{Mx},\ \Delta a \to -10^9\ \text{cm},\ c \to 3\times 10^{10}\ \text{cm/s},\ \mu 0 \to 4\,\pi\,10^{-7}\ \text{H/m}\};$$

Find the value of the parameter λ evaluating the flux function at y=∞

$$\left(\lambda V = N\left[\frac{\psi Src}{\pi\,Lx} \text{ /. values}\right]\right) \text{ // Framed}$$

$$\boxed{3.1831\times 10^{11}\ \text{Mx/cm}}$$

```
AppendTo[values, λ → λV];
```

Solve for the current by equating the change in flux from part b. to that found in part d.

```
Icgs2mks = 10^8 / ((10 H/m) (3 × 10^10 cm/s));
sol3 = NSolve[Δψp12 == Δψ12b /. values, Ics];
IcsV = Re[sol3[[1, 1, 2]]];
AppendTo[values, Ics → IcsV];
UnitConvert[IcsV Icgs2mks, "amps"] // Framed
```

$\boxed{-4.90637 \times 10^{12}\ \text{A}}$

Now we are free to find a value for the length of the current sheet using the results of part d.

```
L /. values // Framed
```

$\boxed{9.39594 \times 10^{9}\ \text{cm}}$

The height of the current sheet was found in part c.

```
N[h /. values] // Framed
```

$\boxed{4.24264 \times 10^{9}\ \text{cm}}$

and the energy was found in part e.

```
(ΔϵMV = UnitConvert[ΔϵM Lx / μ0 /. values, "ergs"]) // Framed
```

$\boxed{-6.74193 \times 10^{31}\ \text{ergs}}$

Part h.

Save the additional numeric values to memory

```
vribV = vrib → 3 km/s ;
Bz0V = Bz0 → 300 G ;

AppendTo[values, vribV];
AppendTo[values, Bz0V];
```

The total reconnection time can be found by equating the change in flux to the product of the magnetic field and the change in area. We then solve the expression for time.

```
(τrx = UnitConvert[Δψ12b/(Bz0 vrib) /. values, "seconds"]) // Framed
```

$\boxed{1111.5\ \text{s}}$

we are then free to compute the average power using the quotient of the total energy release and the total reconnection time

```
(PM = ΔϵMV / τrx) // Framed
```

$\boxed{-6.06561\times 10^{28}\ \text{ergs/s}}$

The electric field is the negative change in flux per unit length

```
(EM = UnitConvert[- Δψ12b / τrx /. values, "volts/meter"]) // Framed
```

$\boxed{-90.\ \text{V/m}}$

The energy flux incident on each ribbon is the power divided by the area of the ribbon

```
energyFlux = PM/(Lx vrib τrx) /. values // Framed
```

$\boxed{-1.81904\times 10^{10}\ \text{ergs/(cm}^2\text{s)}}$

```
AppendTo[values, ρ0 → 10^-15 g/cm^3];
machnum = UnitConvert[τA /. values, "second"] / τrx // Framed
```

$\boxed{0.011553}$

```
τA /. values
```

$0.000406074\ \sqrt{\text{g}}\ \text{cm}\ \sqrt{\text{H}}/\text{Mx}$

```
UnitConvert[τA /. values, "second"]
```

12.8412 s